

Journal of Liquid Chromatography & Related Technologies
Publication details, including instructions for authors and subscription information:
http://www.informaworld.com/smpp/title~content=t713597273

A Microcomputer Program for the Calculation of R_F Values of Solutes in Stepwise Gradient Thin-Layer Chromatography

W. Markowski[a]; E. Soczewiński[a]; G. Matysik[a]
[a] Department of Inorganic and Analytical, Chemistry Medical Academy Staszica 6 20-081, Lublin, Poland

To cite this Article Markowski, W. , Soczewiński, E. and Matysik, G.(1987) 'A Microcomputer Program for the Calculation of R_F Values of Solutes in Stepwise Gradient Thin-Layer Chromatography', Journal of Liquid Chromatography & Related Technologies, 10: 7, 1261 — 1276

To link to this Article: DOI: 10.1080/01483918708066767

URL: http://dx.doi.org/10.1080/01483918708066767

JOURNAL OF LIQUID CHROMATOGRAPHY, 10(7), 1261-1276 (1987)

A MICROCOMPUTER PROGRAM FOR THE CALCULATION OF R_F VALUES OF SOLUTES IN STEPWISE GRADIENT THIN-LAYER CHROMATOGRAPHY

W. Markowski, E. Soczewiński, and G. Matysik

Department of Inorganic and Analytical Chemistry
Medical Academy
Staszica 6
20-081 Lublin, Poland

ABSTRACT

A microcomputer program (in BASIC, e.g., for SPECTRUM ZX+) is proposed for the calculation of final R_F values obtained under conditions of stepwise gradient. After introduction of R_F values of sample components obtained for several isocratic runs, the microcomputer calculates the R_F values for any gradient program.Good agreement of calculated and experimental R_F values was obtained. The stepwise gradient development was carried out using a sandwich chamber with a glass distributor; the eluent fractions according to a chosen stepwise program were introduced directly under the distributor with a micropipette.

INTRODUCTION

In a preceding paper (1) an equation for the R_F value of a solute chromatographed under stepwise gradient conditions was derived, assuming definite relationships between k′ values and modifier concentration:

$$R_F = \sum_{i=1}^{h-1} \frac{v_i}{k'(j,i)} + R_F(j,h)\left[1 - \sum_{i=1}^{h-1} x(j,i)\right]$$

where

$$\sum_{i=1}^{h-1} x(j,i) = \sum_{i=1}^{h-1} \frac{v(i)}{1 - R_F(j,i)}$$

and

$$R_F(j,i) = \frac{1}{1 + k'(j,i)} = \frac{1}{1 + a(j,i)c^{-m}}$$

where:

j - the number of solute

i - the number of elution step (eluent fraction)

h - the number of the last step in which the solute migrates through a part of the concentration zone

k′(j,i) - capacity factor of solute j (i-th step)

$R_F(j,i)$ - corresponding R_F value; simplified symbol r (j,i) is used in the program

v(i) - volume of eluent introduced in the i-th step

c(i) - concentration of modifier in the i-th step

x(j,i) - the volume of mobile phase corresponding to the migration of solute "j" through the i-th concentration zone.

m - slope of linear log k´ vs. log c plot

a(j) - constant (log k´ value for pure modifier, c=1)

For detailed discussion, see (1).

As a reference system, 20 hypothetical solutes of a wide range of capacity factors ($1:2^{19}$) were chosen; the R_F vs. modifier concentration relationships of components of a real sample can be compared to the hypothetical solutes, especially for similar slopes of log k´ vs. log c plots. It is thus possible to study by computer simulation the final arrangement of spots for chosen programs of stepwise gradient.

In the present paper a modified microcomputer program is presented,which simplifies the calculation of R_F values obtained for stepwise gradient elution. The R_F values of individual components obtained for isocratic runs are introduced directly into the microcomputer. Since curvilinear relationships are frequently obtained for log k´ vs. log c or log k´ vs. c plots, the experimental points are fitted to a quadratic function. The program is formulated in BASIC for a popular microcomputer of the Spectrum ZX+ type; the display shows:

table of codes and names of solutes

table of **isocratic** R_F values

log k´ vs. log c plot and the corresponding coefficients of the function $A_o + A_1 \log c + A_2 (\log c)^2$

shape of the gradient program

table of final R_F values; plot representing the migration of solutes as function of eluent volume.

The program is given in Table I

Table I. Microcomputer program (BASIC) for the simulation of stepwise gradient elution. Introduced data: R_F values for given concentrations (c) of the modifier; stepwise gradient program (volumes of eluent fractions relative to the void volume and the corresponding concentrations of modifier, c).

```
  1 LLIST
 1Ø PRINT AT Ø,7; "ISOCRATIC ELUTION"
 2Ø PRINT
 3Ø INPUT "THE POLAR SOLVENT"; p$
 4Ø INPUT "THE DILUENT"; d$
 5Ø INPUT "THE ADSORBENT"; a$
 6Ø PRINT
 65 PRINT p$; "-"; d$; "-"; a$;: PRINT
 7Ø INPUT "THE NUMBER OF SOLUTES b=";b
 8Ø DIM u (b) : DIM x (b,1Ø) : DIM r (b ,1Ø) : DIM p (b,1Ø)
 9Ø DIM n$ (b,2Ø) : DIM c$ (b,5)
 95 DIM k (b,1Ø)
1ØØ DIM a (b) : DIM b (b) : DIM c (b) : DIM d (b) : DIM e (b) :
    DIM f (b)
11Ø DIM g (b) : DIM h (b) : DIM i (b) : DIM j (b) : DIM q (1ØØ) :
    DIM o (b,1ØØ)
12Ø FOR z=1 TO b
13Ø INPUT "THE CODE OF SOLUTE"; c$ (z)
14Ø INPUT "THE NAME OF SOLUTE";n$ (z)
15Ø NEXT z
16Ø LET p=INT (b/14+1)
165 FOR i=1 TO p
17Ø IF b<14 THEN GO TO 2ØØ
18Ø LET k1=14
19Ø GO TO 21Ø
2ØØ LET k1=b
21Ø FOR z=1 TO k1
22Ø PRINT AT 6,1; "z"; AT 6,4; "CODE"
23Ø PRINT AT 6,11; "THE NAME OF SOLUTE"
24Ø PRINT AT (7+z),1;z;AT (7+z) , 4; c$ (z) ; AT (7+z),11;
    n$ (z)
```

```
25Ø PLOT Ø,128: DRAW 255,Ø: PLOT 255,Ø: DRAW Ø,128
26Ø PLOT Ø,Ø: DRAW Ø,128: PLOT Ø,Ø: DRAW Ø,128
27Ø PLOT Ø,112: DRAW 255,Ø: PLOT 23,Ø: DRAW Ø,128
28Ø PLOT 8Ø,Ø: DRAW Ø,128
29Ø NEXT z
3ØØ STOP
31Ø COPY: CLS
32Ø NEXT i
33Ø STOP
34Ø GO SUB 4ØØØ
35Ø PRINT AT 1,1; "z"; AT 1,4; "CODE"; AT 1,12; "c"; AT
    1,22; "Rf"
36Ø LET p=INT (b/3+1)
37Ø FOR i=1 TO p
38Ø LET k1=3* (i-1)+3
39Ø IF k1<b THEN GO TO 41Ø
4ØØ LET k1=b
41Ø FOR z=3*(i-1)+1 TO k1
42Ø IF z>3 THEN GO TO 45Ø
43Ø PRINT AT 3+(z-1)*6,1; z; AT 3+(z-1)*6,4; c$(z)
44Ø GO TO 47Ø
45Ø GO SUB 45ØØ
46Ø GO SUB 5ØØØ
465 PRINT AT (1+(h-1)*6),1; z;AT (1+(h-1)*6),4;c$(z)
47Ø INPUT "THE NUMBER OF EXPERIMENTAL POINTS f="; u(z)
48Ø LET p=u(z)
49Ø FOR l=1 TO p
5ØØ INPUT "THE CONCENTRATION OF MODIFIER c="; x(z,l)
51Ø INPUT "THE VALUES OF Rf="; r(z,l)
52Ø IF z>3 THEN GO TO 55Ø
53Ø PRINT AT (z-1)*6+l+2, 12;x(z,l) ; AT (z-1)*6+l+2,21;
    r(z,l)
54Ø GO TO 57Ø
55Ø GO SUB 5ØØØ
56Ø PRINT AT (h-1)*6+l,12; x(z,l) ; AT(h-1)*6+l,21;
    r(z,l)
57Ø LET x(z,l)=Ø.434*LN x(z,l)
58Ø LET k(z,l)=(1-r(z,l)) /r(z,l)
59Ø LET k(z,l)= Ø.434*LN k(z,l)
6ØØ LET x(z,l)= INT (x(z,l)*1ØØØ)*Ø.ØØ1
61Ø LET k(z,l)=INT (k(z,l)*1ØØØØ)*Ø.ØØØ1
62Ø NEXT l
63Ø IF z<=i*3 THEN GO TO 65Ø
64Ø COPY: CLS
65Ø NEXT z
66Ø COPY: CLS
67Ø NEXT i
68Ø STOP
69Ø GO SUB 55ØØ
7ØØ FOR z=1 TO b
71Ø LET p=u(z)
```

```
72Ø FOR l=1 TO p
73Ø IF k(z,l)<=-2 THEN GO TO 76Ø
74Ø IF k(z,l)>=+2 THEN GO TO 76Ø
745 IF x(z,l)<=-2 THEN GO TO 76Ø
75Ø CIRCLE 248+x(z,l)*246,84+k(z,l)*42,2
76Ø NEXT l
77Ø NEXT z
78Ø/FOR z=1 TO b
79Ø FOR l=1 TO 1ØØ
8ØØ LET q(l)=Ø.434*LN(Ø.1+Ø.ØØ9*(l-1))
81Ø LET o(z,l)=h(z)+q(l)*(i(z)+j(z)*q(l))
82Ø IF o(z,l)<=-2 THEN GO TO 85Ø
83Ø IF o(z,l)>=+2 THEN GO TO 85Ø
84Ø CIRCLE 248+q(l)*246,84+o(z,l)*42,.5
85Ø NEXT l
86Ø NEXT z
87Ø PLOT Ø,Ø: DRAW Ø,168: PLOT Ø,84: DRAW 246,Ø
88Ø PLOT 246,Ø: DRAW Ø,168:
89Ø FOR i=1 TO 4
9ØØ CIRCLE 248,i*42,1
91Ø NEXT i
92Ø FOR i=1 TO 2
93Ø CIRCLE 2+(i-1)*246,84,1
94Ø NEXT i
95Ø PRINT AT Ø,Ø; "log c-> ";AT Ø,27;"Rm=+2"
96Ø PRINT AT 11,31;"Ø";AT 21,27;"Rm=-2";AT 12,Ø;"-1"
98Ø FOR z=1 TO b
985 IF h(z)>=2 OR h(z)<=-2 THEN GO TO 1ØØØ
99Ø PRINT AT 21-(84+h(z)*42)/8.4,31;z
1ØØØ NEXT z
1Ø3Ø STOP:COPY: CLS
1Ø4Ø LPRINT AT 1,7;"STEPWISE GRADIENT"
1Ø5Ø LPRINT
1Ø6Ø INPUT "THE NUMBER OF STEPS n=";n
1Ø7Ø PRINT "THE NUMBER OF STEPS n=";n
1Ø8Ø DIM r(b,n) : DIM c(n) : DIM k(b,n) : DIM x(b,n) :
     DIM y(b,n) : DIM v(n)
1Ø85 DIM s(b,n) : DIM z(b,n)
1Ø9Ø PRINT
11ØØ PRINT "THE CONCENTRATION OF MODIFIER ON i-TH STEP"
111Ø PRINT
112Ø LET v=Ø
113Ø FOR i=1 TO n
114Ø INPUT "c=";c(i),"v=";v(i)
115Ø PRINT "c(";i;")=";c(i),"v(";i;")=";v(i)
116Ø LET v=v+v(i)
117Ø NEXT i
118Ø STOP: CLS
119Ø FOR i=1 TO n
12ØØ LET v(i)= v(i)/v
121Ø PRINT "c(";i;")=";c(i),"v(";i;")=";v(i)
122Ø NEXT i
```

```
123Ø COPY: CLS
124Ø GO SUB 6ØØØ
125Ø STOP
126Ø LPRINT
127Ø LPRINT AT 1,9;"THE S-P SYSTEM"
128Ø LPRINT
129Ø DEF FN k(j,i)=1Ø↑(h(j)+Ø.434×(LN c(i))×(i(j)+
     j(j)×Ø.434×LN c(i)))
13ØØ DEF FN r(j,i)=INT (1ØØØ/(1+FN k(j,i)))×Ø.ØØ1
131Ø LPRINT "THE DISTANCE TRAVELLED BY SPOTS AFTER N
     DEVELOPMENT STEPS"
132Ø LPRINT
133Ø FOR j=1 TO b
134Ø FOR i=1 TO n
135Ø LET x(j,i)=v(i)/(1-FN r(j,i))
136Ø LET y(j,i)=v(i)/FN k(j,i)
137Ø NEXT i
138Ø LET s=Ø
1390 FOR i=1 TO n
14ØØ LET s=s+x(j,i)
141Ø LET s(j,i)=s
142Ø NEXT i
143Ø FOR i=1 TO n
144Ø IF i>=2 AND s(j,i)>=1 THEN GO TO 152Ø
145Ø IF s(j,i)>=1 THEN GO TO 148Ø
146Ø NEXT i
147Ø GO TO 163Ø
148Ø LET r=FN r(j,i)
149Ø PLOT 1Ø,1Ø: DRAW 16Ø,r×16Ø
15ØØ LPRINT "Rf(";c$(j);")=";r
151Ø GO TO 163Ø
152Ø LET z(j,i)=(1-s(j,i-1))×FN r(j,i)
1535 LET r=Ø
154Ø FOR p=1 TO i-1
155Ø LET r=r+y(j,p)
156Ø NEXT p
157Ø LET r(j,i)=INT (1ØØØ×(r+z(j,i)))×Ø.ØØ1
158Ø LPRINT "Rf(";c$(j);")=";r(j,i)
159Ø PLOT 1Ø,1Ø
16ØØ FOR h=1 TO i-1
161Ø DRAW x(j,h)×16Ø,y(j,h)×16Ø
162Ø NEXT h
1625 DRAW (1-s(j,i-1))×16Ø,z(j,i)×16Ø
163Ø NEXT j
164Ø LET v=Ø
165Ø FOR i=1 TO n
166Ø LET v=v+v(i)
167Ø PRINT AT 21,v×2Ø;" ";v
168Ø PLOT 1Ø+v×16Ø,1Ø: DRAW (1-v)×16Ø,(1-v)×16Ø
169Ø CIRCLE 1Ø+i×16Ø/n,1Ø,1
17ØØ CIRCLE 171,1Ø+i×16Ø/n,1
```

```
171Ø NEXT i
172Ø PRINT AT 1Ø,24;"Rf=Ø.5";AT Ø,24;"↑";"Rf";
     AT 21,24;"-->";"V,X,S"
173Ø PLOT 1Ø,1Ø: DRAW 16Ø,Ø: DRAW Ø,16Ø: DRAW -16Ø,O:
     DRAW Ø,-16Ø
1735 DRAW 16Ø,16Ø
174Ø COPY: CLS
175Ø INPUT "REPEAT PROFILE OF GRADIENT a$=1";a$
176Ø IF a$="1" THEN GO TO 1Ø4Ø
177Ø IF a$<> "1" THEN GO TO 1Ø
4ØØØ PLOT, Ø,Ø: DRAW Ø,175: PLOT Ø,175: DRAW 255,Ø
4Ø1Ø PLOT 255,Ø: DRAW Ø,175: PLOT Ø,152: DRAW 255,Ø
4Ø2Ø PLOT 23,Ø: DRAW Ø,175: PLOT 16Ø,Ø: DRAW Ø,175
4Ø3Ø PLOT 8Ø,Ø: DRAW Ø,175:
4Ø4Ø RETURN
45ØØ PLOT Ø,Ø: DRAW Ø,175: PLOT 255,Ø: DRAW Ø,175
451Ø PLOT 23,Ø: DRAW Ø,175: PLOT 8Ø,Ø: DRAW Ø,175
452Ø PLOT 16Ø,Ø: DRAW Ø,175
453Ø RETURN
5ØØØ IF z=3*(i-1)+1 THEN LET h=1
5Ø1Ø IF z=3*(i-1)+2 THEN LET h=2
5Ø2Ø IF z=3*(i-1)+3 THEN LET h=3
5Ø3Ø RETURN
55ØØ FOR i=1 TO b
551Ø LET a(i)=Ø: LET b(i)=Ø: LET c(i)=Ø: LET d(i)=Ø
552Ø LET e(i)=Ø: LET f(i)=Ø: LET g(i)=Ø
553Ø LET p=u(i)
554Ø FOR j=1 TO p
555Ø LET a(i)=a(i)+x(i,j)
556Ø LET b(i)=b(i)+x(i,j)*x(i,j)
557Ø LET c(i)=c(i)+x(i,j)*x(i,j)*x(i,j)
558Ø LET d(i)=d(i)+x(i,j)*x(i,j)*x(i,j)*x(i,j)
559Ø LET e(i)= e(i)+x(i,j)*k(i,j)
56ØØ LET f(i)=f(i)+x(i,j)*x(i,j)*k(i,j)
561Ø LET g(i)=g(i)+k(i,j)
562Ø NEXT j
563Ø LET w1=p*b(i)-a(i)*a(1)
564Ø LET w2=p*f(i)-b(i)*g(i)
565Ø LET w3=p*c(i)-a(i)*b(i)
566Ø LET w4=p*e(i)-a(i)*g(i)
567Ø LET w5=p*d(i)-b(i)*b(i)
568Ø LET j(i)=(w1*w2-w3*w4)/(w1*w5-w3*w3)
569Ø LET i(j)=(w4-j(i)*w3)/w1
57ØØ LET h(i)=(g(i)-j(i)*b(i)-i(i)*a(i))/p
571Ø LET h(i)= INT(h(i)*1ØØØ)*Ø.ØØ1
572Ø LET i(i)= INT(i(i)*1ØØØ)*Ø.ØØ1
573Ø LET j(i)= INT(j(i)*1ØØØ)*Ø.ØØ1
574Ø NEXT i
575Ø PRINT AT 1,1;"CODE";AT 1,1Ø;"Ao";AT 1,18;"A1";
     AT 1,26;"A2"
576Ø PLOT Ø,175: DRAW 255,Ø: PLOT Ø,Ø: DRAW Ø,175
577Ø PLOT 56,Ø: DRAW Ø,175: PLOT 12Ø,Ø: DRAW Ø,175
```

```
5780 PLOT 184,0: DRAW 0,175: PLOT 255,0: DRAW 0,175
5790 PLOT 0,152: DRAW 255,0
5800 FOR i=1 TO b
5810 PRINT AT 3+i,1;c$(i) ;AT 3+i,8;h(i)
5820 PRINT AT 3+i,16;i(i) ;AT 3+i,24;j(i)
5830 NEXT i
5840 STOP: COPY: CLS
5850 RETURN
6000 LPRINT "THE PROFILE OF STEPWISE GRADIENT"
6010 LPRINT
6020 FOR i=1 TO n
6030 IF i>=2 THEN GO TO 6060
6040 PLOT 0,c(i)*160: DRAW 160/n,0
6050 GO TO 6080
6060 PLOT (i-1)*160/n,c(i)*160: DRAW 160/n,0
6070 PLOT (i-1)*160/n,c(i-1)*160: DRAW 0,
     (c(i)-c(i-1))*160
6080 NEXT i
6090 PLOT 0,0: DRAW 0,160: DRAW 160,0:
     DRAW 0,-160: DRAW -160,0
6100 FOR i=1 TO n
6110 PRINT AT 21,(i-1)*20/n;i;AT 21-20*c(i),20;
     "_(";i;")=";c(i)
6120 NEXT i
6130 STOP: COPY: CLS
6140 RETURN
```


Example 1. For solutes A-G the following R_F values were obtained for various volume fraction concentrations of the modifier:

Solute	Concentration of modifier/R_F value		
A	0.01/0.203	0.03/0.697	0.06/0.902
B	0.03/0.365	0.15/0.615	0.07/0.758
C	0.07/0.439	0.1/0.615	0.2/0.864
D	0.1/0.285	0.3/0.782	0.6/0.935
E	0.1/0.09	0.2/0.285	0.4/0.615
F	0.3/0.183	0.5/0.384	0.9/0.669
G	0.5/0.037	0.8/0.09	0.9/0.112

The following stepwise gradient program is introduced:

0.05; 0.1; 0.2; 0.3; 0.4; v = 0.2.



FIGURE 1a. Printout of gradient profile and resulting separation of solutes A-G.
Figure 1 and Figure 2 are copied from original printouts.

The resulting printout is shown in Fig.1a. It can be seen that the R_F values are too low. Next a steeper program is tried: 0.05; 0.1; 0.2; 0.5; 1.0. The resulting printout is shown in Fig.1b.

The separation is now satisfactory: the spots are evenly spaced which corresponds to the best separation and highest capacity for micropreparative separations.



FIGURE 1b. Separation of solutes A-G for modified gradient program.

EXPERIMENTAL

A mixture of phenols and phenolic acids was chromatographed on the distance of 10 cm in the system silica-chloroform + ethyl acetate (modifier). Silica 60 F 254 layers on glass carrier plates were used (5 x 20 cm, E.Merck, Darmstadt, F.R.G.). Equilibrium sandwich chambers for TLC (2-4) produced by Polish Reagents POCh, Lublin were used. In stepwise gradient experiments 0.1 ml fractions of the eluent were introduced directly under the distributor as described in an earlier paper (2) according to the following program: 3x0.1, 3x0.2, 3x0,7, 3x1.0 (volume fractions of ethyl acetate in chloroform, a four-step gradient). The volumes of the eluent fractions, in void volume units, were thus equal to 0.25.

The samples were spotted after prewetting the plate with one distributor volume (0.1 ml) of 10% ethyl acetate in chloroform. The spots were detected in iodine vapours.

RESULTS AND DISCUSSION

The experimental data for isocratic elution runs are presented in Table II.

In Fig.2 the printout obtained after introduction of the gradient program (volume-concentration: 0.25, 0.1; 0.25, 0.2; 0.25, 0.7; 0.25, 1.0) and the isocratic R_F values is presented. Comparison of calculated and

Table II. R_F values of solutes obtained for isocratic elution at various modifier concentrations.

Solute/Code	Concentration of EtOAc, vol.fractions					
	0.1	0.2	0.3	0.5	0.7	1.0
1-naphthol/1HB	0.70	0.775	0.84	0.92	-	-
2-naphthol/2HB	0.61	0.70	0.775	0.86	-	-
Umbelliferon/UMB	0.19	0.33	0.43	0.60	0.73	0.845
Orcinol/ORC	0.14	0.235	0.35	0.60	0.77	0.93
Phloroglucinol/ FLORO	0.00	0.05	0.09	0.27	0.47	0.79
4-nitrophenol/ 4NPH	0.38	0.54	0.60	0.735	0.83	0.91
Gallic Acid/GAC	0.00	0.00	0.00	0.07	0.18	0.43
	(c=0.8, R_F=0.26; c=0.9, R_F=0.35)					

experimental R_F values shows good agreement between the two sets of values. Some minor discrepancies are under standable if the complex elution process is considered and possible distortion of the gradient profile in the system.

As reported earlier, additional improvement of separation in stepwise gradient elution is caused by compression of zones due to the sudden changes of eluent strength and enhancement of mutual displacement effects. Therefore, in spite of deterioration of separation of umbelliferon and orcinol in the third modifier concentration zone (ΔR_F = 0.02), the separation of the two flattened spots was satisfactory.

STEPWISE GRADIENT

c (1) = 0.1	v (1) = 0.25
c (2) = 0.2	v (2) = 0.25
c (3) = 0.7	v (3) = 0.25
c (4) = 1	v (4) = 0.25

THE PROFILE OF STEPWISE GRADIENT



THE S-P SYSTEM

THE DISTANCE TRAVELLED BY SPOTS AFTER N DEVELOPMENT STEPS

Rf(1HB)=0.713	EXP.:	0.69
Rf(2HB)=0.643		0.63
Rf(UMB)=0.414		0.40
Rf(ORC)=0.42		0.42
Rf(FLORO)=0.257		0.30
Rf(4NPH)=0.487		0.54
Rf(GAC)=0.141		0.12



FIGURE 2. Printout of data for phenols and phenolic acids: gradient program and its profile, calculated R_F values (compared to experimental data), movement of spots in the four modifier concentration zones, plot of isocratic data and coefficients of the curves fitted to the experimental points.



CODE	Ao	A1	A2
1HB	-1.688	-2.421	-1.102
2HB	-1.242	-1.696	-0.648
UMB	-0.723	-1.913	-0.573
ORC	-1.080	-3.433	-1.584
FLOR	-0.562	-4.006	-1.966
4NPH	-0.979	-1.885	-0.716
GAC	0.115	-3.570	-0.733

CONCLUSIONS

The simulation of the stepwise gradient process permits the prediction of R_F values from the corresponding isocratic data. The method thus permits the choice of suitable gradient program from a few isocratic data and is thus especially valuable when the sample to be separated by preparative TLC is scarce (e.g., in biochemical or forensic analysis). The microcomputer program can also be utilized for teaching purposes since it illustrates the principle of gradient elution and the effect of the gradient profile on the separation achieved.

REFERENCESS

1. E.Soczewiński and W.Markowski, Stepwise gradient development in thin-layer chromatography. III. A computer program for the simulation of stepwise gradient elution, J.Chromatogr., 370 (1986) 63.

2. G.Matysik and E.Soczewiński, Stepwise gradient development in thin-layer chromatography. II. Two-dimensional gradients for complex mixtures, J.Chromatogr., 369 (1986) 19.

3. E.Soczewiński, Equilibrium sandwich TLC chamber for continuous development with a glass distributor, in R.E.Kaiser (Editor), Planar Chromatography, Vol.1 Hüthig, Heidelberg 1986, p.79.

4. G.Matysik and E.Soczewiński, On-line extraction and preconcentration of solid samples in equilibrium-sandwich chambers for thin-layer chromatography, J.Chromatogr., 355 (1986) 363.